

Classified as Internal / Clasificado como Interno

Sub Abarrotes_2CD200()
    Dim WS1 As Worksheet, a As Integer, old As Integer
    Dim WS2 As Worksheet, b As Integer

    Dim Fil As Integer, Fila As Long, _
    Col As Integer, Total As Long

    Dim Tabla_Col() As Long, Ini As Single
    Dim Tabla_Fil() As String
    Dim Tabla_DAT() As Long, Inicio As Single

    Dim Tabla_Transit() As Long
    Dim Tabla_Reserva() As Long
    Dim Tabla_Retenid() As Long

    Inicio = Timer

    ' ---&--- Limpia la Hoja

    Worksheets("Inv-PedAbarrotes_2").Range("D6:EM75").Value = Empty

    With Application
        .ScreenUpdating = False
        old = .Calculation
        .Calculation = xlCalculationManual
        .EnableEvents = False
    End With

    Set WS1 = Worksheets("Inventario")
    Set WS2 = Worksheets("Inv-PedAbarrotes_2")

    ' ---&--- Carga los datos en las tablas

    ' En primer lugar hay que entender como se comporta Excel: Cuando hay varias celdas combinadas la celda que
    ' tiene los datos es la primera fila (superior) y la primera columna (izquierda).
    ' En este caso están combinadas las columnas D+E+F (4+5+6) la que tiene el texto es la primeria la D.
    ' El primer While recorre la columnas de 3 en 3 para buscar los códigos hasta que encuentre la 
    ' columna vacía.
    ' El Redim redimensiona la tabla y el Preserve hace que conserve los valores anteriores.
    ' Las líneas de código tachadas no sirven em este caso.
    ' El segundo While recorre las fechas, las guarda en una tabla que va redimensionando

    With WS2
        ' ---&--- Tabla Columnas

        Col = 4
        While .Cells(5, Col).Value <> ""
            ReDim Preserve Tabla_Col(Col)
            ReDim Preserve Tabla_Transit(Col)
            ReDim Preserve Tabla_Reserva(Col)
            ReDim Preserve Tabla_Retenid(Col)
            Tabla_Col(Col) = Val(.Cells(4, Col))
            Col = Col + 3
        Wend

        ' ---&--- Tabla Filas

        Fil = 6
        While .Cells(Fil, 3).Value <> ""
            ReDim Preserve Tabla_Fil(Fil)

            Tabla_Fil(Fil) = Format(.Cells(Fil, 3), "yyyy.mm.dd")
            Fil = Fil + 1
        Wend
    End With

   
    ' ---&--- Tabla de datos de las Columnas

    ' Ordena las tablas
    ' Esta parte no es necesaria porque si los códigos de artículos ya están ordenados la tabla esta ordenada.

    For a = 1 To UBound(Tabla_Col) - 1
        For b = a To UBound(Tabla_Col)
            If Tabla_Col(a) > Tabla_Col(b) Then
                Tabla_Col(0) = Tabla_Col(a)
                Tabla_Col(a) = Tabla_Col(b)
                Tabla_Col(b) = Tabla_Col(0)

                Tabla_Transit(0) = Tabla_Transit(a)
                Tabla_Transit(a) = Tabla_Transit(b)
                Tabla_Transit(b) = Tabla_Transit(0)

                Tabla_Reserva(0) = Tabla_Reserva(a)
                Tabla_Reserva(a) = Tabla_Reserva(b)
                Tabla_Reserva(b) = Tabla_Reserva(0)

                Tabla_Retenid(0) = Tabla_Retenid(a)
                Tabla_Retenid(a) = Tabla_Retenid(b)
                Tabla_Retenid(b) = Tabla_Retenid(0)
            End If
        Next
    Next

    ' ---&--- Tabla de datos de las Filas

    For a = 1 To UBound(Tabla_Fil) - 1
        For b = a To UBound(Tabla_Fil)
            If Tabla_Fil(a) > Tabla_Fil(b) Then
                Tabla_Fil(0) = Tabla_Fil(a)
                Tabla_Fil(a) = Tabla_Fil(b)
                Tabla_Fil(b) = Tabla_Fil(0)
            End If
        Next
    Next

    ' ---&--- Tabla de datos

    ' Aquí se crea una tabla que corresponde a las celdas a rellenar de la hoja.
    ' Tiene tantas columnas como columnas tenga de artículos y tantas filas como fechas.

    ' Por ejemplo el articulo 310101 devuelve la columna 4 – Col=4.
    '   El valor de Invent ira a Tabla_Dat(Fil, Col+0)
    '   El valor de Orden  ira a Tabla_Dat(Fil, Col+1)
    '   El valor de Asigna ira a Tabla_Dat(Fil, Col+2)

    ' Por ejemplo el articulo 310102 devuelve la columna 7 – Col=7.
    '   El valor de Invent ira a Tabla_Dat(Fil, Col+0)
    '   El valor de Orden  ira a Tabla_Dat(Fil, Col+1)
    '   El valor de Asigna ira a Tabla_Dat(Fil, Col+2)

    ReDim Tabla_DAT(Fil, Col)

    ' ---&--- Lee la hoja de datos

    With WS1
        Fila = 2
        While .Cells(Fila, "A") <> ""

            If (UCase(.Cells(Fila, "A")) = "200VR" Or .Cells(Fila, "A") = 200) And _
                UCase(.Cells(Fila, "O")) = "PLF" Then

                Col = Buscar_2(.Cells(Fila, "C"), Tabla_Col)
                Fil = Buscar_2(Format(.Cells(Fila, "I"), "yyyy.mm.dd"), Tabla_Fil)

                Select Case UCase(.Cells(Fila, "P"))
                    Case "STOCK LIB": Tabla_DAT(Fil, Col) = Tabla_DAT(Fil, Col) + .Cells(Fila, "G")
                    Case "TRANSITO": Tabla_Transit(Col) = Tabla_Transit(Col) + .Cells(Fila, "G")
                    Case "RESERVA": Tabla_Reserva(Col) = Tabla_Reserva(Col) + .Cells(Fila, "G")
                    Case "RETENIDO": Tabla_Retenid(Col) = Tabla_Retenid(Col) + .Cells(Fila, "G")
                End Select
            End If

            Fila = Fila + 1
        Wend
    End With

    ' ---&--- Escribe los datos en la hoja Refrigerado

    With WS2
         For Col = 4 To UBound(Tabla_Col)
             For Fil = 6 To UBound(Tabla_Fil)
                 If Tabla_DAT(Fil, Col) <> 0 Then
                     .Cells(Fil, Col) = Tabla_DAT(Fil, Col)
                     Total = Total + Tabla_DAT(Fil, Col)
                 End If
             Next

             If Col = 4 Then
                 .Cells(Fil + 3, "B") = "Transito"
                 .Cells(Fil + 4, "B") = "Reserva"
                 .Cells(Fil + 5, "B") = "Retenido"
                 .Cells(Fil + 1, "B") = "Total Cajas"
             End If

             If Tabla_Transit(Col) <> 0 Then .Cells(Fil + 3, Col) = Tabla_Transit(Col)
             If Tabla_Reserva(Col) <> 0 Then .Cells(Fil + 4, Col) = Tabla_Reserva(Col)
             If Tabla_Retenid(Col) <> 0 Then .Cells(Fil + 5, Col) = Tabla_Retenid(Col)
             .Cells(Fil + 1, Col) = Total
             Total = 0#
         Next
    End With
    WS2.Cells(1, 12) = "INFORME DE ABA CD SANTIAGO"
    MsgBox "Fin de la Macro Ver. 2.00." & vbCrLf & "Tiempo: " & Timer - Inicio
End Sub



    ' Busca en una tabla ordenada.
    ' El funcionamiento es el siguiente: 
    ' - se coloca en la mitad de la tabla y pueden pasar tres cosas:
    '   - Que sea el valor buscado
    '   - Que sea menor en este caso salta a la mitad de la primera mitad
    '   - Que sea mayor en este caso salta a la mitad de la segunda mitad
    ' – Se repite el ciclo cada vez saltando a una mitad más pequeña.

    ' En una tabla con 1.048.576 elementos examinando como máximo 20 elementos puedo saber si existe o no existe
    ' El elemento buscado y no es necesario recorrer toda la tabla.

Function Buscar_2(Texto, Tabla)
    Dim Maxim As Integer, Punte As Integer, Mitad As Single, Tipo As String

    Maxim = UBound(Tabla)
    Punte = 2 ^ Int((Log(Maxim) / Log(2)) + 0.99)
    Mitad = Punte

    While Mitad > 0.5
        Mitad = Mitad / 2
        If Punte > Maxim Then
            Tipo = "MAYOR"
        Else
            If Tabla(Punte) = Texto Then Tipo = "IGUAL"
            If Tabla(Punte) < Texto Then Tipo = "MENOR"
            If Tabla(Punte) > Texto Then Tipo = "MAYOR"
        End If

        Select Case Tipo
             Case "IGUAL"
                 Buscar_2 = Punte
                 Mitad = 0

             Case "MENOR"
                 Punte = Punte + Mitad

             Case "MAYOR"
                 Punte = Punte - Mitad
        End Select
    Wend
End Function



